
function NextPrime(n)
    (n<=2)    && return 2
    (n%2==0)  && return NextPrime(n+1)
    PrimeP(n) && return n
    NextPrime(n+2)
end

function SolvayStrassen(n)
    n==2 && return true
    n>2 && (n%2==zero(n)) && return false
    W=(n+3)÷2
    for a in 2:W
        if (powermod(a,(n-1)÷2,n)!=(n+Jacobi(a,n))%n) 
            print("Witness=",a,"\n")
            return false
        end
    end
    true
end

function SPprime(n,a)
    t,s = n-1,0
    while (t%2==0)
        s +=  1
        t >>= 1
    end
    b=powermod(a,t,n)
    ((b==1) | (b==n-1)) && return true
    for j in 1:s-1
        b=powermod(b,2,n)
        b==n-1 && return true
    end
    false
end

function Miller(n)
    W = Integer(min(round(2*log(n)^2),n-1))
    for a in 2:W
        SPprime(n,a) || return false
    end
    true
end



function AKS(n)
    
    "Approximates (as integer) the k-th root of n"
    function int_root(n,k)
        x,y=1<<Integer(ceil(msb(n)/k)),0
        while(y<x)
            (y>0) && (x=y)
            y=((k-1)*x+(n÷x^(k-1)))÷k
        end
        x
    end
    
    "Is n an i-th power?"
    function ispower(n,i)
        int_root(n,i)^i==n
    end
        
    "Is n a perfect power for some i?"
    function ispower(n)
        for i=2:Integer(ceil(log2(n)))
            ispower(n,i) && return true
        end
        return false
    end

    "Find an integer r such that Or(n)>log2(n)^2"
    function findOrder(n)
        lval=Integer(floor(log2(n))^2)+1
        uval=Integer(floor(max(3,log2(n)^5)))
        for r=lval:uval
            (1∉[powermod(n,i,r) for i in 1:lval]) && return r
        end
    end

    "Compute (A*B  mod (X-1)^r) mod n where r=lenght(B)"
    function ⋆(A::Vector{T},B::Vector{T}) where T<: Integer
        r=length(B)
        Rpol=fill(0,r)
        for i in 0:r-1
            Rpol+=A[i+1]*circshift(B,i)
            Rpol.%=n
        end
        Rpol
    end
        
 
    "Compute (X+a)^n mod (X-1)^r) mod n"
    function Ξ(a::Integer,n::Integer,r::Integer)
        Xpol=fill(0,r+1)
        Xpol[[1,2]]=[a,1]
        Rpol=fill(0,r+1)
        Rpol[1]=1
        for i in msb(n):-1:0
            Rpol=Rpol⋆Rpol
            (((n>>i) & 1)==1) && ( Rpol=Rpol⋆Xpol )
        end
        Rpol
    end
    # Step 1



    # Step 1
    ispower(n) && return false
    # Step 2
    r=findOrder(n)
    # Step 3
    for a in 2:r
        gcd(a,n)!=1 && gcd(a,n)!=n && return false
    end
    # Step 4
    n≤r && return true
    # Step 5
    for a in 1:Integer(ceil(√(r))*ceil(log2(n)))
        Xpol=Ξ(a,n,r)
        Xpol[1]-=a
        Xpol[(n%r)+1]-=1
        Xpol = Xpol .% n
        Xpol==0*Xpol || return false 
    end
    # Step 6
    return true
end

using Plots
plotlyjs()

function iscollision(x)
    length(unique(x))<length(x)
end

function samplebirtday(n,i,tr=5000)
    j=0
    for _ in 1:tr
        iscollision(rand(1:n,i)) && (j+=1)
    end
    j/tr
end

function actualest(n,i)
    1-reduce(*,[1-j/n for j in 0:i-1])
end

function expest(n,i)
    C=√(-2*log(1-1/2))
    v=1-C*exp(-i*(i+1)/(2n))
    v>0 ? v : 0
end

function getprobs(n,tr=5000)
    R=[ 0 0 0 ]
    for i in 1:n
        R=[R ; [samplebirtday(n,i,tr) actualest(n,i) expest(n,i)] ]
    end
    R
end

function plotprobs(n,tr=5000)
    cv=ceil(√(2log(2)*n))
    m=getprobs(n,tr)
    plot(xrange=(0,n),yrange=(0,1))
    hline!([1/2],label="")
    vline!([cv,cv+2],label="")
    plot!(m[:,1],label="Sample")
    plot!(m[:,2],label="Actual")
    plot!(m[:,3],label="Exp")
end

